

**L i s t i n g :**    CODEC (Programm zum Ver- und Entschluessein von Dateien)

**100** ' \*\*\*\*\*

**110** ' \* \*

**120** ' \* *DATEI-CODIERER & DECODIERER* \*

**130** ' \* \*

**140** ' \*\*\*\*\*

```

150 '
160 MODE 1:INK 0,13:INK 1,0:INK 2,0,3:INK 3,1
170 PAPER 0:PEN 3:BORDER 10:SPEED INK 30,20
180 SYMBOL AFTER 250
190 SYMBOL 254,&66,&0,&66,&66,&66,&66,&3E,&0
200 SYMBOL 255,&66,&0,&3C,&66,&66,&66,&3C,&0
210 LOCATE 1,1:PRINT STRING$(40,210);
220 LOCATE 1,2
230 PRINT CHR$(24)+"          DATEI-CODIERER & DECODIERER          "+CHR$(24)
240 LOCATE 10,5:PRINT CHR$(164);" Martin Kotulla 1985"
250 PRINT:PRINT STRING$(40,210):PRINT
260 PEN 1:PRINT " Mit diesem Programm k"+CHR$(255)+"nnen Sie Dateien"
270 PRINT:PRINT " nach einem beliebigen Schl"+CHR$(254)+"ssel"
280 PRINT:PRINT " codieren und wieder decodieren."
290 PEN 3:PRINT:PRINT STRING$(40,210):PRINT:PEN 1
300 LOCATE 1,20:PEN 2
310 WHILE INKEY$<>"" :WEND ' Tastaturpuffer loeschen
320 PRINT " Bitte dr"+CHR$(254)+"cken Sie eine Taste!"
330 WHILE INKEY$="" :WEND ' Auf Tastendruck warten
340 MODE 1:PEN 3
350 PRINT:PRINT " Bitte geben Sie den Code-Schl"+CHR$(254)+"ssel an!"
360 PEN 1:PRINT:PRINT STRING$(40,210):PRINT
370 PEN 3:PRINT " ";:LINE INPUT code$
380 PEN 1:PRINT:PRINT STRING$(40,210):PRINT:PEN 3
390 PRINT " Name der Eingabedatei:":PEN 1:PRINT:LINE INPUT " ",infile$
400 PEN 3:PRINT:PRINT " Name der Ausgabedatei:":PEN 1
410 PEN 1:PRINT:LINE INPUT " ",outfile$
420 OPENIN infile$
430 OPENOUT outfile$
440 PEN 1:PRINT:PRINT STRING$(40,210):PEN 3
450 code$=UPPER$(code$)
460 IF LEN(codier$)>254 THEN 480
470 codier$=codier$+MID$(code$,1,255-LEN(codier$)):GOTO 460
480 PRINT " [1] Codieren"
490 PRINT:PRINT " [2] Decodieren"
500 cd$=UPPER$(INKEY$):IF cd$="" THEN 500
510 IF cd$="1" THEN 550
520 IF cd$="2" THEN 650
530 PRINT CHR$(7);:GOTO 500
540 ' CODIERUNG *****
550 WHILE NOT EOF
560   b$="":LINE INPUT #9,a$
570   FOR i=1 TO LEN(a$)
580     b$=b$+CHR$((ASC(MID$(a$,i)) XOR ASC(MID$(codier$,i))) OR 128)
590   NEXT i
600   PRINT #9,b$
610 WEND
620 CLOSEIN:CLOSEOUT:CLEAR
630 MODE 1:PEN 1:END
640 ' DECODIERUNG *****
650 WHILE NOT EOF
660   b$="":LINE INPUT #9,a$
670   FOR i=1 TO LEN(a$)
680     b$=b$+CHR$((ASC(MID$(a$,i)) XOR ASC(MID$(codier$,i))) AND 127)
690   NEXT i
700   PRINT #9,b$
710 WEND
720 CLOSEIN:CLOSEOUT:CLEAR
730 MODE 1:PEN 1:END
740 ' *****

```